NAG Fortran Library Routine Document

G01FBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01FBF returns the deviate associated with the given tail probability of Student's t-distribution with real degrees of freedom, via the routine name.

2 Specification

```
real FUNCTION G01FBF(TAIL, P, DF, IFAIL)
INTEGER              IFAIL
real                 P, DF
CHARACTER*1          TAIL
```

3 Description

The deviate, t_p associated with the lower tail probability, p, of the Student's t-distribution with ν degrees of freedom is defined as the solution to

$$P(T < t_p : \nu) = p = \frac{\Gamma((\nu+1)/2)}{\sqrt{\nu\pi}\Gamma(\nu/2)} \int_{-\infty}^{t_p} \left(1 + \frac{T^2}{\nu}\right)^{-(\nu+1)/2} dT, \quad \nu \geq 1;\ -\infty < t_p < \infty.$$

For $\nu = 1$ or 2 the integral equation is easily solved for t_p.

For other values of $\nu < 3$ a transformation to the beta distribution is used and the result obtained from G01FEF.

For $\nu \geq 3$ an inverse asymptotic expansion of Cornish–Fisher type is used. The algorithm is described by Hill (1970).

4 References

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

Hill G W (1970) Student's t-distribution *Comm. ACM* **13** 617–619

5 Parameters

1: TAIL – CHARACTER*1 *Input*

On entry: indicates which tail the supplied probability represents.

If TAIL = 'U', the upper tail probability, i.e., P($T \geq t_p : \nu$).

If TAIL = 'S', the two-tail (significance level) probability, i.e., P($T \geq |t_p| : \nu$) + P($T \leq -|t_p| : \nu$).

If TAIL = 'C', the two-tail (confidence interval) probability, i.e., P($T \leq |t_p| : \nu$) − $P(T \leq -|t_p| : \nu)$.

If TAIL = 'L', the lower tail probability, i.e., P($T \leq t_p : \nu$).

Constraint: TAIL = 'U', 'S', 'C' or 'L'.

2: P – ***real*** *Input*

On entry: the probability, p, from the required Student's t-distribution as defined by TAIL.

Constraint: $0.0 < \mathrm{P} < 1.0$.

3: DF – ***real*** *Input*

On entry: the degrees of freedom, ν, of the Student's t-distribution.

Constraint: $\mathrm{DF} \geq 1.0$.

4: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: $\mathrm{IFAIL} = 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if $\mathrm{IFAIL} \neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $\mathrm{IFAIL} = 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

If $\mathrm{IFAIL} = 1$, 2, 3 or 4 on exit, then G01FBF returns zero.

$\mathrm{IFAIL} = 1$

On entry, TAIL $\neq$ 'U', 'S', 'C' or 'L'.

$\mathrm{IFAIL} = 2$

On entry, $\mathrm{P} \leq 0.0$,
or $\mathrm{P} \geq 1.0$.

$\mathrm{IFAIL} = 3$

On entry, $\mathrm{DF} < 1.0$.

$\mathrm{IFAIL} = 4$

The solution is too close to zero to be determined accurately. This error will only occur when $\mathrm{DF} = 1.0$. The returned value of zero will be a good approximation in terms of absolute value but will have a poor relative precision.

$\mathrm{IFAIL} = 5$

Convergence in the calculation of the inverse beta value was not achieved. However, the result should be a reasonable approximation to the correct value.

7 Accuracy

The results should be accurate to 5 significant digits, for most parameter values. The error behaviour for various parameter values is discussed in Hill (1970).

8 Further Comments

The value t_p may be calculated by using the transformation described in Section 3 and using G01FEF. This routine allows the user to set the required accuracy.

9 Example

Lower tail probabilities are read for several t-distributions, and the corresponding deviates calculated and printed, until the end of data is reached.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G01FBF Example Program Text
*     Mark 14 Release.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             DF, P, X
      INTEGER          IFAIL
      CHARACTER        TAIL
*     .. External Functions ..
      real             G01FBF
      EXTERNAL         G01FBF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G01FBF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      WRITE (NOUT,*)
      WRITE (NOUT,*) '     P      DF    TAIL      X'
      WRITE (NOUT,*)
   20 READ (NIN,*,END=40) P, DF, TAIL
      IFAIL = -1
*
      X = G01FBF(TAIL,P,DF,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
         WRITE (NOUT,99999) P, DF, TAIL, X
      ELSE
         WRITE (NOUT,99999) P, DF, TAIL, X, ' NOTE: IFAIL = ', IFAIL
      END IF
      GO TO 20
   40 STOP
*
99999 FORMAT (1X,2F8.3,3X,A1,3X,F8.3,A,I1)
      END
```

9.2 Program Data

```
G01FBF Example Program Data
0.0100  20.0  'S'         :P DF TAIL
0.01     7.5  'L'         :P DF TAIL
0.99    45.0  'C'         :P DF TAIL
```

9.3 Program Results

```
 G01FBF Example Program Results

      P      DF    TAIL      X

    0.010  20.000   S      2.845
    0.010   7.500   L     -2.943
    0.990  45.000   C      2.690
```